Всем привет. Столкнулся с такой проблемой: при нажатии на кнопку "Заполнить показатели" строки позиций товаров меняются местами (см. скриншоты). Вопрос в том, как сделать так, чтобы этого не происходило? Пока не могу понять.

Прилагаю код самой команды:

```1c
    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | СебестоимостьКВыгрузкеПроекты.Сделка КАК Сделка,
    | &Организация КАК Организация
    |ПОМЕСТИТЬ ВТ_СебестоимостьКВыгрузке
    |ИЗ
    | &ТЗ КАК СебестоимостьКВыгрузкеПроекты
    |;
    |
    |////////////////////////////////////////////////////////////­////////////////////
    |ВЫБРАТЬ
    | ДополнительныеРасходыПоСделке.Регистратор,
    | СУММА(ЕСТЬNULL(ДополнительныеРасходыПоСделке.Сумма, 0)) КАК Сумма,
    | ЕСТЬNULL(ДополнительныеРасходыПоСделке.Регистратор.ДатаВходящегоДокумента, "") КАК ДатаВходящегоДокумента,
    | ЕСТЬNULL(ДополнительныеРасходыПоСделке.Регистратор.НомерВходящегоДокумента, "") КАК НомерВходящегоДокумента,
    | ВТ_СебестоимостьКВыгрузке.Сделка,
    | ВТ_СебестоимостьКВыгрузке.Организация
    |ПОМЕСТИТЬ ВТ_ДопРасходыПоСделкеРасшифровка
    |ИЗ
    | ВТ_СебестоимостьКВыгрузке КАК ВТ_СебестоимостьКВыгрузке
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления.ДополнительныеРасходыПоСделке КАК ДополнительныеРасходыПоСделке
    | ПО ВТ_СебестоимостьКВыгрузке.Организация = ДополнительныеРасходыПоСделке.Организация
    | И ВТ_СебестоимостьКВыгрузке.Сделка = ДополнительныеРасходыПоСделке.Сделка
    |
    |СГРУППИРОВАТЬ ПО
    | ДополнительныеРасходыПоСделке.Регистратор,
    | ВТ_СебестоимостьКВыгрузке.Сделка,
    | ВТ_СебестоимостьКВыгрузке.Организация,
    | ЕСТЬNULL(ДополнительныеРасходыПоСделке.Регистратор.ДатаВходящегоДокумента, ""),
    | ЕСТЬNULL(ДополнительныеРасходыПоСделке.Регистратор.НомерВходящегоДокумента, "")
    |;
    |
    |////////////////////////////////////////////////////////////­////////////////////
    |ВЫБРАТЬ
    | ВТ_ДопРасходыПоСделкеРасшифровка.Организация,
    | ВТ_ДопРасходыПоСделкеРасшифровка.Сделка,
    | СУММА(ВТ_ДопРасходыПоСделкеРасшифровка.Сумма) КАК Сумма
    |ПОМЕСТИТЬ ВТ_ДопРасходыПоСделке
    |ИЗ
    | ВТ_ДопРасходыПоСделкеРасшифровка КАК ВТ_ДопРасходыПоСделкеРасшифровка
    |
    |СГРУППИРОВАТЬ ПО
    | ВТ_ДопРасходыПоСделкеРасшифровка.Организация,
    | ВТ_ДопРасходыПоСделкеРасшифровка.Сделка
    |;
    |
    |////////////////////////////////////////////////////////////­////////////////////
    |ВЫБРАТЬ
    | ВТ_СебестоимостьКВыгрузке.Сделка КАК Сделка,
    | СУММА(ЕСТЬNULL(ВыручкаИСебестоимостьПродажОбороты.СебестоимостьОборот, 0)) КАК Себестоимость,
    | СУММА(ЕСТЬNULL(ВыручкаИСебестоимостьПродажОбороты.СуммаДополнительныхРасходовОборот, 0)) КАК ДопРасходы,
    | СУММА(ЕСТЬNULL(ВТ_ДопРасходыПоСделке.Сумма, 0)) КАК РасходыПоСделке,
    | "" КАК Описание
    |ИЗ
    | ВТ_СебестоимостьКВыгрузке КАК ВТ_СебестоимостьКВыгрузке
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления.ВыручкаИСебестоимостьПродаж.Обороты КАК ВыручкаИСебестоимостьПродажОбороты
    | ПО (ВТ_СебестоимостьКВыгрузке.Сделка = ВЫРАЗИТЬ(ВыручкаИСебестоимостьПродажОбороты.ЗаказКлиента КАК Документ.ЗаказКлиента).Сделка)
    | И ВТ_СебестоимостьКВыгрузке.Организация = ВыручкаИСебестоимостьПродажОбороты.АналитикаУчетаПоПартнерам.Организация
    | ЛЕВОЕ СОЕДИНЕНИЕ ВТ_ДопРасходыПоСделке КАК ВТ_ДопРасходыПоСделке
    | ПО ВТ_СебестоимостьКВыгрузке.Организация = ВТ_ДопРасходыПоСделке.Организация
    | И ВТ_СебестоимостьКВыгрузке.Сделка = ВТ_ДопРасходыПоСделке.Сделка
    |
    |СГРУППИРОВАТЬ ПО
    | ВТ_СебестоимостьКВыгрузке.Сделка
    |;
    |
    |////////////////////////////////////////////////////////////­////////////////////
    |ВЫБРАТЬ
    | ВТ_ДопРасходыПоСделкеРасшифровка.Регистратор,
    | ВТ_ДопРасходыПоСделкеРасшифровка.Сумма,
    | ВТ_ДопРасходыПоСделкеРасшифровка.Сделка,
    | ВТ_ДопРасходыПоСделкеРасшифровка.Организация,
    | ВТ_ДопРасходыПоСделкеРасшифровка.ДатаВходящегоДокумента,
    | ВТ_ДопРасходыПоСделкеРасшифровка.НомерВходящегоДокумента
    |ИЗ
    | ВТ_ДопРасходыПоСделкеРасшифровка КАК ВТ_ДопРасходыПоСделкеРасшифровка";

    Запрос.УстановитьПараметр("Организация", Объект.Организация);
    Запрос.УстановитьПараметр("ТЗ", Объект.Проекты.Выгрузить());

    Результат = Запрос.ВыполнитьПакет();
    ВыборкаДопРасходы = Результат[4].Выбрать();

    // Заполним таблицу проектов
    Объект.Проекты.Загрузить(Результат[3].Выгрузить());

    // Дополним таблицу проектов данными по доп. расходам (сведения о документах - Описание)
    Для Каждого Элемент Из Объект.Проекты Цикл

    ВыборкаДопРасходы.Сбросить();
    Отбор = Новый Структура("Сделка", Элемент.Сделка);

    Пока ВыборкаДопРасходы.НайтиСледующий(Отбор) Цикл

    Если ПустаяСтрока(ВыборкаДопРасходы.НомерВходящегоДокумента) И ПустаяСтрока(ВыборкаДопРасходы.ДатаВходящегоДокумента) Тогда
    ПредставлениеДокумента = Строка(ВыборкаДопРасходы.Регистратор);
    Иначе
    ПредставлениеДокумента = ВыборкаДопРасходы.НомерВходящегоДокумента + " от " + Формат(ВыборкаДопРасходы.ДатаВходящегоДокумента, "ДФ=dd.MM.yyyy");
    КонецЕсли;

    Элемент.Описание = Элемент.Описание + ?(ПустаяСтрока(Элемент.Описание), ", "; ") + ПредставлениеДокумента;
    КонецЦикла;

    КонецЦикла;
```
---
Необходимо сделать:

```1c
    ВЫБРАТЬ
    | СебестоимостьКВыгрузкеПроекты.НомерСтроки,
    | СебестоимостьКВыгрузкеПроекты.Сделка КАК Сделка,
    | &Организация КАК Организация
    |ПОМЕСТИТЬ ВТ_СебестоимостьКВыгрузке
    |ИЗ
    | &ТЗ КАК СебестоимостьКВыгрузкеПроекты
```

И в остальных запросах пакета так же использовать поле НомерСтроки.
В 4-ом запросе пакета использовать Упорядочить ПО НомерСтроки
---
Необходимо сделать:

```1c
    ВЫБРАТЬ
    | СебестоимостьКВыгрузкеПроекты.НомерСтроки,
    | СебестоимостьКВыгрузкеПроекты.Сделка КАК Сделка,
    | &Организация КАК Организация
    |ПОМЕСТИТЬ ВТ_СебестоимостьКВыгрузке
    |ИЗ
    | &ТЗ КАК СебестоимостьКВыгрузкеПроекты
```

И в остальных запросах пакета так же использовать поле НомерСтроки.
В 4-ом запросе пакета использовать Упорядочить ПО НомерСтроки
---
Сейчас попробую
---
Спасибо, работает!
---
И как я понял, документы становятся в порядке их создания (на первом месте самый ранний, на последнем самый поздний)
---
Так надо смотреть команду. Наверное, в ней сортировка какая-то есть.
---
Это код всей команды